UNPUBLISHED PRELIMINARY DATA

N64 28836

(ACCESSION NUMBER)

90

(PAGES)

NASA CR 58608

(NASA CR OR TMX OR AD NUMBER)

1

(THRU)

(CODE)

15

(CATEGORY)

FACILITY FORM 602

OTS PRICE

XEROX $ 8.10ph

MICROFILM $

UNIVERSITY OF MARYLAND
COMPUTER SCIENCE CENTER

COLLEGE PARK, MARYLAND

5 5

REPORTS CONTROL No.

Technical Report TR64-10 August 1964

NsG-398

Pattern Recognition: II. RAMP and REST

Two Image Analysis Programs

by

James N. Orton
Azriel Rosenfeld

The computer programs described in this report were written under the supervision of Dr. Azriel Rosenfeld, Research Associate Professor (P. T.) at the Computer Science Center. Programming personnel were Messrs. James N. Orton, Ernest W. Smith and Bernard Altschuler of the Budd Information Sciences Center, McLean, Virginia. Their time was supported by Contract NAS5-3461 between The Budd Company and the National Aeronautics and Space Administration. The computer time used was supported almost entirely by the National Aeronautics and Space Administration under Grant NSG-398 to the University of Maryland.

ABSTRACT

This report presents detailed descriptions of two basic computer programs for the analysis of the connectivities and shapes of regions in digitized images. The application of these programs to the analysis of TIROS cloud cover pictures and outline maps made from such pictures is also briefly described.

The programs are written in FAP. They have been extensively tested on the 7090.

Author

CONTENTS

LIST OF FIGURES

1. Introduction

Computer programs for processing pictorial information have received some attention in recent years. However, research in this field has tended to be concentrated in specific areas, among them

(a) Development of algorithms for "local" processing of two-dimensional data arrays, with application to "smoothing" or otherwise simplifying given pictorial information (Dineen 1955, Kirsch 1957, Unger 1958, Narasimhan 1964)

(b) Development of techniques for processing pictures which consist of discrete, well-defined parts in given topological and metric relationships (Freeman 1961, Hodes 1961, Sutherland 1963, Kirsch 1964)

This report describes two computer programs designed as basic tools for analyzing the sizes and shapes of connected regions in an image defined by those image points having a given range of brightness (or density) values. These programs represent one approach to bridging the gap in digital picture processing research between local processing on the one hand and the processing of pictures consisting of discrete parts on the other.

The first of these programs, called the Region Area Measurement Program (RAMP), identifies, computes the area of, and displays connected regions on a given set of digitized pictures input from magnetic tape.

As a by-product it also produces from the line-by-line, element-by-element digital representation of the pictures a condensed representation on magnetic tape in terms of the picture-element intervals belonging to each region on succeeding lines. Each interval is tagged by a number uniquely identifying the region of which it is a part. This tape is thus a reconstructible condensation of the information displayed (printed out) by RAMP. It may be used as input to other image analysis programs, for example the second program described in this report.

The second program, called the Region-Enclosed Square Tabulator (REST), produces from this interval representation of a picture a frequency distribution of the squares of all possible sizes enclosed within each region. Two square types are distinguished: these which touch a region boundary ("exterior" squares) and these which do not ("interior" squares).

RAMP and REST have been developed as general-purpose tools for analyzing the connected regions on any image. RAMP distinguishes the regions and measures their sizes. REST provides basic information about the shapes of the regions; for example, a distribution of squares which falls off abruptly for large square sizes is intuitively indicative of an elongated region.

The next two sections of this report describe in detail the structure and operation of the RAMP and REST programs. The concluding section discusses the application of both programs to representative TIROS cloud cover pictures and to outline drawings (nephanalyses) made from such pictures.

2. RAMP: Region Area Measurement Program

2.1 General description

The principal function of RAMP is to identify and measure the connected regions defined by points having a given range of brightness values on a digitized picture. Two important examples of pictures to which RAMP can be applied are

(a) Any digitized picture, where the given range of brightness values are those exceeding an arbitrary threshold. An example is the TIROS picture shown in Figure 1, where the brightness threshold may be selected (Arking 1964) so as to discriminate between the cloud and noncloud portions of the picture. (For this threshold, RAMP will analyze the connected masses of cloud cover on the picture.)

(b) A digitized line drawing consisting of black lines on a uniform white background. For this type of picture and a reasonable choice of brightness threshold, RAMP will analyze the regions enclosed by the lines. An example is the set of twenty-eight cloud cover maps (made from TIROS pictures) shown in Figure 2.

The input to the program is a magnetic tape digital representation of the picture or set of pictures to be analyzed. This tape can be prepared from an original hard copy picture by scanning it as described in the next subsection.



Figure 1

Typical TIROS Cloud Picture



FIGURE 2.
TYPICAL CLOUD COVER MAPS

The program produces three outputs:

(a) A pictorial printout as illustrated in Figure 3, corresponding to the upper left corner picture in Figure 2, in which each connected region of above-threshold picture elements has been assigned a distinguishing symbol. The set of symbols to be used for this purpose can be specified by the user. Up to forty-seven characters can be used. In the printout, each element of a connected region is represented by the character assigned to the region. The below-threshold elements are represented by blanks. The printout is thus a "negative" in the sense that "dark" regions are output as blanks, while "bright" regions are printed out as aggregates of characters.

(b) A printed list of the connected regions and their areas.* Such a list for the picture of Figure 3 is shown in Figure 4.

(c) A magnetic tape condensation representing the pictures in terms of the intervals in which a set of parallel scan lines intersect each region. These intervals are expressed in the form (a,b), where a and b are the picture element numbers of the leftmost and rightmost elements in the given interval. The elements are numbered left to right from 1 to P, the picture width. (A detailed description of the format of this tape, which is the input to the REST program, is given in Section 3.)

*The area of a region is defined as the number of picture elements it contains.



FIGURE 3

RAMP PRINTOUT

FOR A TYPICAL MAP

Region	Region Area
Y	180
1	140
2	4
U	7410
3	597
4	456
5	15
6	694
7	160
Z	1066

Figure 4

LIST OF REGIONS IN FIGURE 3

2.2 Input

The images used as input to RAMP were converted into an IBM-compatible digital tape using a high resolution optical-mechanical scanner currently in operation at the National Bureau of Standards. (The continuing cooperation of Mr. Leonard Cahn of the Bureau in making this scanner available is gratefully acknowledged.) This scanner was made by converting a metal lathe. The image to be scanned is placed on a three inch diameter drum which is inserted in the lathe. At one end of the drum there is a slip ring which can be rotated with respect to the drum. On the slip ring is cemented a photograph of white and black lines and a black space corresponding to clock pulses and an inter-record gap. The rotation of the slip ring with respect to the drum determines the point at which the scanner starts to scan the photograph.

The optical system used in the scanner consists of a light source, a lens, a square aperture (1/192" x 1/192"), and a multiplier phototube. Electrical signals are obtained at the output of the multiplier phototube. A second optical system is placed in front of the slip ring and the output of its multiplier phototube provides clock pulses for the entire electrical system. This second optical system is connected to the automatic feed of the lathe. Rotation of the drum provides the lateral motion across the picture and the automatic feed of the lathe provides the longitudinal motion.

The output of the main multiplier phototube is fed into a video amplifier and amplified. The amplified signal is then fed into a three bit analog-to-digital converter. The resulting binary signal is then fed into an Ampex digital tape recorder and recorded on magnetic tape at 200 bits per inch.

The scanner is equipped with preselected start and stop limits. The maximum over-all scanning area is 9 x 10.4 inches. The area is scanned from left to right starting in the upper left hand corner. Each line is equal to one record; its end is marked by an end of record gap. Each record contains 288 words representing 1728 picture elements. The entire scanning area contains approximately 2000 records.

2.3 Operating parameters

RAMP is coded for an IBM 7090 computer in the FAP symbolic assembly language. For input-output a University of Maryland modified version of the FORTRAN I/O Hollerith package was used.

Parameter cards must be supplied for each run to specify the number and size of the pictures in the sequence to be processed. All pictures to be processed during a single run of the program must be stored on a single magnetic tape. Each record on the tape contains information about one picture line; lines are numbered consecutively from 1 to L, where L is the total number of lines stored on the tape.

The parameter cards to be supplied for a set of p pictures are the following:

Symbolic Location and Contents			Parameter Definition	Parameter Limits
ENDR + 0	DEC	R	Picture 1:	1 to magnetic tape storage limit
			. number of lines in picture	
+ 1	DEC	W_b	. beginning word of line (6 picture elements per word)	$1 \leq W_b \leq W_e$
+ 2	DEC	W_e	. ending word of line	$W_b \leq W_e \leq 20$
ENDR + 3, 4, 5			Picture 2: same	
. . .				
ENDR + 3(p-1) + 0, 1, 2			Picture p: same	
ENDR + 3p	DEC	0	Signal for end of picture sequence	

These cards are inserted in the symbolic program deck just prior to the last symbolic card ("END" card).

2.4 Logical description

The input to RAMP is a binary tape containing digitized information from a scanned image. The scanner described in Subsection 2.2 assigns each map element to one of eight gray levels; the program, however, is concerned only with "black" (below threshold) and "white" (above threshold) elements.

One row (= line) of elements is read in at a time. For it, the program records all of the intervals between "black" elements, giving their beginning and ending element numbers. For the first row, these intervals are assigned successive region numbers and a count is started for each region giving the number of elements in the interval. Starting with the second row, all of the intervals for a given row m are compared with the intervals of the previous row m-1 in the following manner: If an interval (c, d) of row m overlaps an interval (a, b) of row m-1 (that is, if c is not greater than b + 1 of if d is not less than a - 1) then the interval (c, d) is tagged with the region number of the interval (a, b), and the number of elements in (c, d) is added to the cumulative count of that region. If an interval of row m overlaps two intervals (a_1, b_1) and (a_2, b_2) of row m-1, the cumulative count for the region of the interval (a_2, b_2) is added to the cumulative/ count for the region (a_1, b_1) and the region number of (a_2, b_2) is eliminated, (a_2, b_2) being tagged with the region number of (a_1, b_1). If an interval (c, d) of row m does not overlap with any interval in row m-1, then it is assigned a new region number. Each time processing for a line is completed, its tagged intervals are written on magnetic tape.

It is necessary to make only one pass over all the rows of the picture tape using this parallel type of processing.* Furthermore, at any given time only one row of elements and the intervals from two rows are in memory, thus minimizing the amount of storage required.

*However, one additional pass over the interval tape is required to complete the numbering of the regions.

For each region the program puts out the coordinates of an identifying element and the total number of elements. The identifying element is the first element to which the program had assigned that region number which it eventually assigns to the entire region after all "overlaps" have been consolidated as described above. If a region has no "overlaps," this is simply the first element of the region encountered by the scan (the upper left element). For a "U" shaped region, the identifying element will be the upper left element of the left arm of the "U", even though the left arm is shorter than the right arm, so that the right arm was encountered first.

The symbols identifying the regions for output are assigned in rotating order A, B, ..., Z, 1, 2, ..., 9, +, -, /, =, ', .,), $, *, "comma", (corresponding to the order in which the scan encountered the upper left element of the entire region (rather than the identifying element). However, the regions are listed in the output in order of their identifying elements. This explains the order of listing of the region data in Figure 4, above, and Figure 7, below.

Figures 5 and 6 show a detailed flow chart and the 7090 symbolic assembly (FAP) listing for RAMP.

Figure 5

RAMP Flow Chart



Figure 5 - Sheet 1



Figure 5 - Sheet 2



Figure 5 - Sheet 3



DUPLICATES PROCESSING



Figure 5 - Sheet 4



Figure 5 - Sheet 5



Figure 5 - Sheet 6



Figure 5 - Sheet 7



Figure 5 - Sheet 8

Figure 6

RAMP Symbolic Program Listing

```
*                   RAMP    REGION AREA MEASUREMENT PROGRAM
 X1        CLA      BEGW
           SUB      ONE
           STO      BEGW
           CLA      ENDW
           ADD      P0+2
           STA      A1+1
           STA      A21+1
           STA      P5-1
           STA      P5-2
           STA      P5+2
           STA      Q3+3
           STA      Q5-3
           STA      Q5+6
           STA      Q8-2
           CLA      ENDW
           SUB      BEGW
           STA      A0+1
           STA      P0
           STA      Q5+3
           LXA      ENDR,1
           SXD      D3-4,1
 X0        CALL     RDSBIN            READ BINARY
           TIX      0,0,8
           TIX      L(IOC),7,0
           TIX      0,1,0
           TRA      A0                READ IN ONE RECORD STARTING
L(IOC)     IORT     GRID,0,-1         IN LOCATION GRID
 A0        AXT      0,4                    A0 THRU A2-1 SEARCHES FOR 1ST WHITE
           AXT      **,2
           AXT      6,1
           CAL      MASK
 A1        ARS      0
           ANA      **,2
           LGR      33
           RQL      1
           LGL      2
           CAS      THREE
           TRA      A2                     TRA IF A WHITE ELEMENT
           NOP
           CLA      A1
           ADD      SIX
           STO      A1
           CLA      LFS
           ADD      ONE
           STO      LFS
           CLA      MASK1
           ANA      A1+2
           SUB      SIX
           STA      A1+2
 A11       TIX      A1-1,1,1               NEXT INST FOR A NEW WORD
           SXA      A1,0
```

Figure 6 - Sheet 1

```
         CLA     TTHREE
         STA     A1+2
         TIX     A1-2,2,1
         TRA     A5A
A2       CLA     CON4                A2 THRU A5-1 SEARCHES FOR NEXT
         STA     A2-1                BLACK ELEM.
         CLA     LFS
         STO     RTS
         CLA     A1
         ADD     SIX
         STA     A21
         CLA     MASK1
         ANA     A1+2
         SUB     SIX
         STA     A21+2
         SXA     A1,0                RESET ROUTINE A1 FOR NEXT ROW
         CLA     TTHREE
         STA     A1+2
         TNX     A4,1,1              TRA FOR NEW WORD
         CAL     MASK
A21      ARS     **
         ANA     **,2
         LGR     **
         RQL     1
         LGL     2
         CAS     THREE              IS THIS ELEM BLACK
         TRA     A3                  =WHITE ELEM.
         NOP
         TRA     A6                  =BLACK ELEM.
A3       CLA     A21
         ADD     SIX
         STO     A21
         CLA     MASK1
         ANA     A21+2
         SUB     SIX
         STA     A21+2
         CLA     RTS
         ADD     ONE
         STO     RTS
         TIX     A21-1,1,1           NEXT INST FOR NEW WORD
A4       SXA     A21,0
         CLA     TTHREE
         STA     A21+2
         AXT     6,1
         TIX     A21-1,2,1
A5       CLA     RTS
         LGR     12
         CLA     LFS
         LGL     12
         STO     CRI,4               LAST INTERVAL IN ROW TO CRI
A501     ZET     LINES            IS THIS THE FIRST ROW
         TXI     B1,4,-1             NOT FIRST ROW
```

Figure 6 - Sheet 2

```
        AXT     0,2                 ASSIGN REG. NOS. TO ALL CRI
        TXI     *+1,4,-1
        SXD     A52+2,4
        PXA     ,4
        PAC     ,4
        SXA     LENG,4
A51     CLA     REGNS
        ADD     ONE
        STO     REGNS
        ALS     24
        ORA     CRI,2               REG. NO. TO CRI
        SLW     PRI,2                    CRI TO PRI
        TSX     XYCOOR,1
        CLA     CON
        ADD     REGNS                    =EFFECTIVE ADDRESS FOR REG SIZE
        STA     A52
        CAL     MASK2
        ANA     PRI,2
        ARS     12
        SLW     TEMP
        CLA     MASK1
        ANA     PRI,2
        SUB     TEMP
        ADD     ONE
A52     STO     **                       SIZE OF CRI TO TABLE REGCT
        TXI     *+1,2,-1
        TXH     A51,2,**
        CLA     ONE
        STO     LINES
        STO     LFS
        STZ     PRI,2                    END OF TABLE PRI
        TRA     X0
A5A     CLA     CON9                     THIS INST. WHEN ROW IS ALL BOL
        STO     D2
        TRA     D1
        CLA     CON2
        STO     D2
        STZ     LENG
        STZ     PRI
        STZ     PRI+1
        CLA     TTHREE
        STA     A1+2
        SXA     A1,0
        TRA     D2+9
A6      CLA     RTS
        LGR     12
        CLA     LFS
        LGL     12
        STO     CRI,4                    LFS,RTS TO CRI
        CLA     RTS
        ADD     TWO
        STO     LFS
        CLA     A21
```

Figure 6 - Sheet 3

```
        ADD     SIX
        STA     A1
        CLA     MASK1
        ANA     A21+2
        SUB     SIX
        STA     A1+2
        CLA     CON3
        STA     A2-1
        TXI     A11,4,-1            GO TO FIND BEGINNING OF NEXT INTERVAL
A7      CLA     CON4
        STA     A2-1
        TXI     A501,4,1            GO TO END OF ROW ROUTINE
B1      AXT     0,1
        AXT     0,2
        CAL     PRI,1
        TZE     C2
        CAL     MASK1               1. IF C GRTR THAN B+1 GET NEXT PRI
        ANA     PRI,1               2. IF D LESS THAN A-1 GET NEXT CRI
        ADD     ONE                 3. IF NEITHER 1 NOR 2 THEN CRI AND
        LGR     36                     PRI OVERLAP
        CAL     MASK2                  (A,B DESIGNATE INTERVAL FOR PRI AND
        ANA     CRI,2                  C,D FOR CRI)
        ARS     12
        TLQ     C1                  TRA IF C GRTR THAN B+1
        CLA     MASK1
        ANA     CRI,2
        LGR     36
        CAL     MASK2
        ANA     PRI,1
        ARS     12
        SUB     ONE
        TLQ     C2                  TRA IF D LESS THAN A-1
        CAL     MASK3               CRI AND PRI OVERLAP
        ANA     CRI,2
        TNZ     B3              TRA IF CRI HAS A REG. NO.
        CAL     MASK3
        ANA     PRI,1
        ORS     CRI,2               REG = OF PRI TO CRI
        ARS     24
        ADD     CON
        STA     B2
        STA     B2+1
        CAL     MASK2
        ANA     CRI,2
        ARS     12
        STO     TEMP
        CLA     MASK1
        ANA     CRI,2
        SUB     TEMP
        ADD     ONE
B2      ADD     **                  SIZE OF CRI + ACCUMULATED SIZE OF
        STO     **              REG. OF PRI TO (REGCT+REG.NO. OF CRI)
        TRA     B5
```

Figure 6 - Sheet 4

```
B3      SLW     TEMP          B3 THRU B5-1 ELIMINATES REDUNDANT REG NO.   1
        CAL     MASK3                                                     2
        ANA     PRI,1                                                     3
        LAS     TEMP                                                      4
        TRA     *+2                                                       5
        TRA     B5           DO NOT ELIMINATE IF REG NOS. ARE EQUAL       6
        CAL     MASK3                                                     7
        ANA     PRI,1
        SLW     TEMP
        ARS     24
        ADD     CON
        STA     B31+1
        CAL     MASK3
        ANA     CRI,2
        ARS     24
        ADD     CON
        STA     B31
        STA     B31+2
B31     CLA     **                REG SIZE OF REG OF PRI + REG
        ADD     **                SIZE OF REG OF CRI TO REG SIZE
        STO     **                OF CRI
        SXA     STOR,1
        AXT     0,1
B32     CAL     MASK3
        ANA     PRI,1
        TZE     B33               ZEROES INDICATE END OF TABLE PRI
        LAS     TEMP
        TXI     B32,1,-1      NOT EQUAL TO REG. NO. OF REDUNDANT REG.
        TRA     *+2           =REG. NO. OF REDUNDANT REG.
        TXI     B32,1,-1          NOT EQUAL
        CAL     MASK1
        ADD     MASK2
        ANS     PRI,1
        CAL     MASK3
        ANA     CRI,2
        ORS     PRI,1         REG. NO. OF CRI TO PRI
        TXI     B32,1,-1
B33     PXA     ,2                                                        1
        TZE     B34+1                                                     2
        SXD     B34,2                                                     3
        AXT     0,1
        CAL     MASK3
        ANA     CRI,1
        LAS     TEMP
        TXI     B34,1,-1
        TRA     *+2           = REG. NO. OF REDUNDANT REG.
        TXI     B34,1,-1
        CAL     MASK1
        ADD     MASK2
        ANS     CRI,1
        CAL     MASK3
        ANA     CRI,2
        ORS     CRI,1         REPLACE REDUNDANT REG. NO.
```

Figure 6 - Sheet 5

```
        TXI     *+1,1,-1
B34     TXH     B33+4,1,**
        LXA     STOR,1
        CLA     TEMP
        ARS     24
        ADD     CON
        STA     *+1
        STZ     **                  CLEAR REG SIZE OF REDUNDANT REG
        CLA     DUPS
        ADD     ONE
        STO     DUPS
        ADD     CON6
        STA     *+5
        CLA     MASK3
        ANA     CRI,2
        ARS     24
        ADD     TEMP
        STO     **
B5      CAL     MASK1
        ANA     PRI,1
        STO     TEMP
        CAL     MASK1
        ANA     CRI,2
        CAS     TEMP
        TRA     C1                  AC GRTR THAN TEMP, GET NEXT PRI
        TXI     *+5,1,-1            AC=TEMP, GET NEXT PRI AND CRI
        TXI     *+1,2,-1            AC LESS THAN TEMP, GET NEXT CRI
        SXD     *+1,4
        TXL     D1,2,**             TRA IF NO MORE CRI
        TRA     B1+2
        CAL     PRI,1
        TNZ     *-5                 TRA IF MORE PRI
        TXI     *-6,1,1
C1      TXI     *+1,1,-1
        CAL     PRI,1
        TNZ     B1+2                TRA IF MORE PRI
        TXI     *+1,1,1
C2      CAL     MASK3
        ANA     CRI,2
        TNZ     C3              TRA IF CRI HAS A REG. NO.
        CLA     REGNS
        ADD     ONE
        STO     REGNS
        ALS     24
        ORS     CRI,2           REG. NO. TO CRI
        ARS     24
        ADD     CON
        STA     C21
        SXA     TEMP,1
        TSX     XYCOOR,1
        LXA     TEMP,1
        CAL     MASK2
        ANA     CRI,2
```

Figure 6 - Sheet 6

```
        ARS     12
        STO     TEMP
        CLA     MASK1
        ANA     CRI,2
        SUB     TEMP
        ADD     ONE
C21     STO     **                  SIZE OF CRI TO (REGCT+REG.NO. OF CRI)
C3      TXI     *+1,2,-1
        SXD     *+1,4
        TXL     D1,2,**                   TRA IF NO MORE CRI
        TRA     B1+2
D1      CAL     =2B17
        CALL    (STB)                     INTERVALS TO TAPE
        SXA     TEMP,2
        AXT     0,2
        LDQ     LENG                      VECTOR LENGTH
        STR
BCK     LDQ     PRI,2
        STR
        TXI     *+1,2,-1
        CLA     PRI,2
        TNZ     BCK
        CALL    (WLR)
        LXA     TEMP,2
        STZ     PRI,2
D2      NOP
        PXA     ,2
        PAC     ,2
        SXA     LENG,2
        PAX     ,2
        TXI     *+1,2,1
        CLA     CRI,2
        STO     PRI,2
        TXH     *-3,2,0
        CLA     LINES
        ADD     ONE
        STO     LINES
        LXA     LINES,1
        TXH     D3,1,**
        CLA     ONE
        STO     LFS
        TRA     X0                        GO TO READ IN NEXT ROW
D3      CAL     =6B17
        CALL    (STH)
        PZE     FORM,0,1
        AXT     8,1
D4      LDQ     HEAD+8,1
        STR                               HEADING TO OP
        TIX     D4,1,1
        STZ     TEMP
        AXT     -1,1
D5      CLA     TEMP
        ADD     ONE
```

Figure 6 - Sheet 7

```
        STO     TEMP
        CLA     REGCT,1
        TZE     D6
        LDQ     TEMP
        STR                         REG NO. TO OP
        CLA     MASK1
        ANA     XYCRD,1
        XCA
        STR                         ROW NO. TO OP
        CLA     XYCRD,1
        ARS     12
        XCA
        STR                         COL NO. TO OP
        LDQ     REGCT,1
        STR                         REG SIZE TO OP
 D6     CLA     REGNS
        CAS     TEMP
        TXI     D5,1,-1
        NOP
        CALL    (FIL)
        CALL    REWTP
        PZE     Z7
        CALL    REWTP
        PZE     ZZ1
        TRA     P0-3
 ZZ1    PZE     0,0,8
 ZZ     PZE     0,0,2
 THREE  OCT     3
 DUPS   OCT     0
        BSS     100
 CON6   STO     DUPS
XYCOOR  CLA     CON5
        ADD     REGNS
        STA     XYR
        STA     XYC
        CLA     LINES
        ADD     ONE
 XYR    STO     **                  ROW(Y COORDINATE) TO XYCRD
        CLA     MASK2
        ANA     CRI,2
 XYC    ORS     **                  COL(X COORDINATE) TO XYCRD
        TRA     1,1
 ONE    OCT     1
 TWO    OCT     2
 SIX    OCT     6
TTHREE  DEC     33
 MASK   OCT     130000000000
 MASK1  OCT     000000007777
 MASK2  OCT     000077770000
 MASK3  OCT     777700000000
 CON    STO     REGCT
 CON2   NOP
 CON3   PZE     A7
```

Figure 6 - Sheet 8

```
CON4   PZE     A5A
LINES  OCT     0
REGNS  OCT     0
LFS    OCT     1
RTS    BSS     1
TEMP   BSS     1
STOR   BSS     1
PRI    BSS     100             TABLE OF PREVIOUS ROW INTERVALS
CRI    BSS     100             TABLE OF CURRENT ROW INTERVALS
GRID   BSS     288
REGCT  BSS     500             TABLE OF ACCUMULATED REGION COUNTS
LENG   PZE     0
CON5   STO     XYCRD
XYCRD  BSS     500             TABLE OF X-Y COORDINATES
FORM   BCI     5,(8A6//(1J12,1J12,1J4,1J16))
HEAD   BCI     5,1  REGION NUMBER      ROW  COL
       BCI     3,     REGION SIZE
SEV    OCT     7
       AXT     500,1
       STZ     XYCRD+500,1    CLEAR TABLE XYCRD
       TIX     *-1,1,1
P0     AXT     **,1
       CLA     CON7
       STO     GRID
       CAL     =2B17
       CALL    (TSB)
       STR
       XCA           1ST WORD CONTAINS NO. OF WORDS IN RECORD
       STO     LENG
       TZE     Q10
       PAX     ,4
       PAX     ,2
       ADD     BCK
       STA     P1                   ='PRI' + NO. OF WORDS IN RECORD
       STA     Q6
       STA     P4
       STA     P6
       STA     *+2
       STR
       STQ     **,4
       TIX     *-2,4,1
       CALL    (RLR)
P01    LXA     DUPS,4
       CLA     CON6
       ADD     DUPS
       ADD     ONE
       STA     P1+3
       CLA     MASK3
P1     ANA     **,2                 EXTRACT REG. NO.
       STO     TEMP
       CLA     MASK3
       ANA     **,4          EXTRACT DUP
       CAS     TEMP
```

Figure 6 - Sheet 9

```
        TRA     *+2
        TRA     Q1                  TRA IF REG. NO. IS A DUP
        TIX     P1+2,4,1
        CAL     TEMP
        ARS     24
        ADD     CON5
        STA     P5                  =EFFECTIVE ADDRESS OF SYMBOL ASSIGNED
        STA     P3                  TO REG. NO.
        STA     *+1
        CAL     **
        TNZ     P4-1                TRA IF REG. HAS A SYMBOL
        CAL     MASK4
P2      ARS     0
        ANA     SYM                 GET NEW SYMBOL FOR THIS REG.
        ALS     0
P3      SLW     **
        CLA     MASK1
        ANA     P2
        ADD     SIX
        LDQ     TTHREE
        TLQ     Q2                  TRA IF NEW WORD OF SYM IS NEEDED
P31     STA     P2
        STA     P3-1
        CLA     COUNT
        ADD     ONE
        STO     COUNT
        LDQ     FFIV
        TLQ     Q9                  TRA IF ALL SYMS. HAVE BEEN USED
        CLA     MASK2
P4      ANA     **,2                EXTRACT LFS OF CURRENT INTERVAL
        ARS     12
        LDQ     ELEM
        TLQ     Q4                  TRA IF CURRENT COL IS LESS THAN LFS
        CAL     MASK4
P41     ARS     0
        COM
        ANA     **,1
        SLW     **,1
P5      CAL     **
        ARS     0
        ORS     **,1                SYMBOL TO GRID
        CLA     MASK1
        ANA     P41
        ADD     SIX
        LDQ     TTHREE
        TLQ     Q3                  TRA IF NEW WORD OF GRID IS NEEDED
P51     STA     P41
        STA     P5+1
        CLA     ELEM
        ADD     ONE
        STO     ELEM
        CLA     MASK1
P6      ANA     **,2                EXTRACT RTS OF INTERVAL
```

Figure 6 - Sheet 10

```
        CAS     ELEM
        NOP
        TRA     P41-1          TRA IF END OF INTERVAL NOT YET REACHED
        TIX     P01,2,1
        TRA     Q7             ALL INTERVALS IN ROW ARE PROCESSED
Q1      CLA     MASK1          REPLACE DUP REG. NO.
        ANA*    P1+3           EXTRACT REPLACING REG. NO.
        LDQ*    P1
        RQL     12
        LGL     24
        SLW*    P1
        TRA     P01
Q2      CLA     MASK5        GET NEXT WORD OF SYMBOLS
        ANA     P2+1
        ADD     ONE
        STA     P2+1
        PXA     ,0
        TRA     P31
Q3      TIX     *+2,1,1
        TRA     Q5             TRA IF ROW IS COMPLETED
        CLA     CON7
        STO     **,1
        PXA     ,0
        TRA     P51
Q4      CLA     ELEM
        ADD     ONE
        STO     ELEM
        CLA     MASK1
        ANA     P41
        ADD     SIX
        LDQ     TTHREE
        TLQ     *+4
Q41     STA     P41
        STA     P5+1
        TRA     P4-1
        TIX     *+2,1,1
        TRA     Q5
        CLA     CON7
        STO     **,1
        PXA     ,0
        TRA     Q41
Q5      CAL     =6B17          ROW IS COMPLETED
        CALL    (STH)
        PZE     FRMT,0,1                                          1
        AXT     **,1
        LDQ     CON7
        STR
        LDQ     **,1
        STR                    ROW J OF GRID TO TAPE
        TIX     *-2,1,1
        CALL    (FIL)
        CAL     =3B17
        CALL    (STB)
```

Figure 6 - Sheet 11

```
         LDQ      LENG
         STR                            ROW J OF INTERVALS TO TAPE
         LXA      LENG,1
Q6       LDQ      **,1
         STR
         TIX      *-2,1,1
         CALL     (WLR)
         CLA      ROW
         ADD      ONE
         STO      ROW
         LDQ      ROW
         CLA      ENDR
         TLQ      Q8
         CALL     REWTP
         PZE      ZZ
         CALL     RUNTP                 REWIND TAPE ON UNIT 3
         PZE      ZZ2
         TRA      Q11
ZZ2      PZE      0,0,3
Q7       TIX      *+2,1,1     REST OF THE ROW IS BOUNDARY
         TRA      Q5
         CLA      CON7
         STO      **,1
         TRA      Q7
Q8       CLA      ONE
         STO      ELEM
         PXA      ,0
         STA      P41
         STA      P5+1
         TRA      P0                    RETURN FOR NEXT ROW
Q9       CLA      CON8
         STA      P2+1
         STZ      COUNT
         PXA      ,0
         STA      P2
         STA      P3-1
         TRA      P4-1
Q10      CALL     (RLR)
         TRA      Q7
CON7     BCI      1,
CON8     PZE      SYM
CON9     TRA      A5A+3
MASK4    OCT      770000000000
ELEM     OCT      1
FFIV     DEC      45
COUNT    OCT      0
ROW      OCT      0
SYM      BCI      8,ABCDEFGHIJKLMNOPQRSTUVWXYZ123456789+-/=',)$*,(
FRMT     BCI      2,(1H9,20A6)
MASK5    OCT      000000077777
Q11      LXA      CON10,1
         TXI      *+1,1,-3
         SXA      CON10,1
```

Figure 6 - Sheet 12

```
         CLA     ENDR,1                                                    4
         TZE     Q12          TRA IF NO MORE PASSES REQUIRED               2
         STO     ENDR                                                      6
         CLA     BEGW,1
         STO     BEGW
         CLA     ENDW,1
         STO     ENDW
         CLA     ONE
         STO     LFS
         STO     ELEM
         STZ     LINES
         STZ     REGNS
         STZ     COUNT
         STZ     ROW
         CLA     CON8
         STA     P2+1
         STZ     DUPS
         PXA     ,0
         STA     P2
         STA     P3-1
         STA     P41
         STA     P5+1
         TRA     X1
Q12      CALL    (SPH)        PRINT ON-LINE
         PZE     FMT,0,1
         AXT     5,1
         LDQ     TAPE+5,1
         STR
         TIX     *-2,1,1
         HPR
         CALL    EXIT
FMT      BCI     1,(5A6)
CON10    DEC     0
TAPE     BCI     5,END OF JOB - DISMOUNT TAPES
ENDR     DEC     289          ENDING ROW
BEGW     DEC     1            BEGINNING WORD
ENDW     DEC     19           ENDING WORD
         DEC     289          EACH SET OF 3 CARDS WIKL CAUSE THE
         DEC     62           PROGRAM TO MAKE ONE PASS ON A  PICTURE.
         DEC     80           THE LAST SET OF 3 CARDS MUST BE FOLLOWED
         DEC     0            BY A WORD OF ZEROES.
         END
```

3. REST: Region-Enclosed Square Tabulator

3.1 General description

The principal function of REST is to compile a frequency distribution of the squares of various sizes lying within each connected region (as defined for RAMP) on a set of pictures. The range of desired sizes, the picture size, and the number of pictures to be processed are specified by the user. These dimensions are all specified in terms of numbers of picture elements.

For each square size considered within a given region, the following information is computed in addition to the square frequency count:

- The frequency of exterior squares (with at least one element touching the boundary of the region)
- The frequency of interior squares (with no element touching the boundary)
- The percentage of squares which are exterior
- The percentage of squares which are interior

Picture data are input to the program from magnetic tape, line by line. Regions are represented by the picture-element intervals which they occupy on each line; a number assigned to each interval indicates its region. The picture interval data are produced by RAMP from an element-by-element digital picture representation as described in Section 2.

A sample picture consisting of ten regions, in the form produced as output by RAMP, is shown in Figure 7. A line drawing representation of this picture is shown in Figure 2, column 1, picture 6. The output of REST for this picture is shown in Figure 8.

3.2 Input

The number of records on the magnetic input tape is equal to the total number of lines in the set of pictures to be processed. Each record contains information about one line in the picture.

The first word of the record contains the number of intervals on the line. Each succeeding word, representing one interval, is divided into three parts of four octal digits each. The first part is the interval region number; the second, the interval left end element number; the third, interval right end element number.

Elements are numbered consecutively from 1 to P, where P is the picture width.

For example, the input record representing line 11 of the sample picture (Figure 7) is:

000000000005 014300010007 013300130050 013400540114 013501210130 013401330162

The region represented by M in the picture was assigned identifying number 143 (octal); by J, identifying number 133; etc.

3.3 Operating parameters

REST is coded in the FAP assembly language to run on an IBM 7090 computer. For input-output a University of Maryland modified FORTRAN Input-Output-Hollerith program was used.



FIGURE 7
TYPICAL INPUT TO REST

Region	Square Width	Square Frequencies Total	Exterior	Interior	Square Percentages Exterior	Interior
L	2	190	111	79	58	42
	3	131	89	42	68	32
	4	79	56	23	71	29
	5	42	32	10	76	24
	6	23	21	2	91	9
	7	10	10	0	100	0
	8	2	2	0	100	0
	9	0	0	0	0	0
J	2	1022	208	814	20	80
	3	913	187	726	20	80
	4	814	172	642	21	79
	5	726	164	562	23	77
	6	642	156	486	24	76
	7	562	148	414	26	74
	8	486	139	347	29	71
	9	414	131	283	32	68
	10	347	123	224	35	65
	11	283	113	170	40	60
	12	224	103	121	46	54
	13	170	96	74	56	44
	14	121	78	43	64	36
	15	74	56	18	76	24
	16	43	40	3	93	7
	17	18	18	0	100	0
	18	3	3	0	100	0
	19	0	0	0	0	0

Figure 8

Output of REST for the Input of Figure 7

Region	Square Width	Total	Square Frequencies Exterior	Interior	Square Percentages Exterior	Interior
K	2	3072	402	2670	13	87
	3	2869	395	2474	14	86
	4	2670	388	2282	15	85
	5	2474	380	2094	15	85
	6	2282	370	1912	16	84
	7	2094	360	1734	17	83
	8	1912	352	1560	18	82
	9	1734	343	1391	20	80
	10	1560	335	1225	21	79
	11	1391	326	1065	23	77
	12	1225	317	908	26	74
	13	1065	310	755	29	71
	14	908	284	624	31	69
	15	755	250	505	33	67
	16	624	228	396	37	63
	17	505	196	309	39	61
	18	396	143	253	36	64
	19	309	103	206	33	67
	20	253	89	164	35	65
	21	206	80	126	39	61
	22	164	72	92	44	56
	23	126	64	62	51	49
	24	92	56	36	61	39
	25	62	46	16	74	26
	26	36	33	3	92	8
	27	16	16	0	100	0
	28	3	3	0	100	0
	29	0	0	0	0	0
O	2	413	112	301	27	73
	3	355	100	255	28	72
	4	301	88	213	29	71
	5	255	79	176	31	69
	6	213	71	142	33	67
	7	176	65	111	37	63
	8	142	59	83	42	58
	9	111	53	58	48	52
	10	83	46	37	55	45
	11	58	39	19	67	33
	12	37	32	5	86	14
	13	19	19	0	100	0
	14	5	5	0	100	0
	15	0	0	0	0	0
P	2	144	67	47	59	41
	3	76	48	28	63	37
	4	47	34	13	72	28
	5	28	25	3	89	11
	6	13	13	0	100	0
	7	3	3	0	100	0
	8	0	0	0	0	0

Figure 8 Sheet 2

Region	Square Width	Total	Square Frequencies Exterior	Interior	Square Percentages Exterior	Interior
N	2	1531	237	1294	15	85
	3	1410	228	1182	16	84
	4	1294	220	1074	17	83
	5	1182	211	971	18	82
	6	1074	200	874	19	81
	7	971	188	783	19	81
	8	874	176	698	20	80
	9	783	164	619	21	79
	10	698	154	544	22	78
	11	619	146	473	24	76
	12	544	138	406	25	75
	13	473	130	343	27	73
	14	406	122	284	30	70
	15	343	115	228	34	66
	16	284	107	177	38	62
	17	228	100	128	44	56
	18	177	94	83	53	47
	19	128	85	43	66	34
	20	83	69	14	83	17
	21	43	40	3	93	7
	22	14	14	0	100	0
	23	3	3	0	100	0
	24	0	0	0	0	0
Q	2	85	76	9	89	11
	3	40	40	0	100	0
	4	9	9	0	100	0
	5	0	0	0	0	0

Figure 8 - Sheet 3

Region	Square Width	Square Frequencies Total	Exterior	Interior	Square Percentages Exterior	Interior
R	2	1388	199	1189	14	86
	3	1286	189	1097	15	85
	4	1189	179	1010	15	85
	5	1097	171	926	16	84
	6	1010	164	846	16	84
	7	926	156	770	17	83
	8	846	149	697	18	82
	9	770	143	627	19	81
	10	697	137	560	20	80
	11	627	129	498	21	79
	12	560	120	440	21	79
	13	498	113	385	23	77
	14	440	106	334	24	76
	15	385	100	285	26	74
	16	334	94	240	28	72
	17	285	88	197	31	69
	18	240	83	157	35	65
	19	197	76	121	39	61
	20	157	69	88	44	56
	21	121	64	57	53	47
	22	88	59	29	67	33
	23	57	51	6	89	11
	24	29	29	0	100	0
	25	6	6	0	100	0
	26	0	0	0	0	0

Figure 8 - Sheet 4

Region	Square Width	Total	Square Frequencies Exterior	Square Frequencies Interior	Square Percentages Exterior	Square Percentages Interior
S	2	14	14	0	100	0
	3	4	4	0	100	0
	4	0	0	0	0	0
M	2	5343	802	4541	15	85
	3	4928	750	4178	15	85
	4	4541	711	3830	16	84
	5	4178	685	3493	16	84
	6	3830	637	3193	17	83
	7	3493	583	2910	17	83
	8	3193	549	2644	17	83
	9	2910	518	2392	18	82
	10	2644	480	2164	18	82
	11	2392	445	1947	19	81
	12	2164	427	1737	20	80
	13	1947	412	1535	21	79
	14	1737	396	1341	23	77
	15	1535	377	1158	25	75
	16	1341	359	982	27	73
	17	1158	345	813	30	70
	18	982	324	658	33	67
	19	813	296	517	36	64
	20	658	256	402	39	61
	21	517	215	302	42	58
	22	402	188	214	47	53
	23	302	160	142	53	47
	24	214	124	90	58	42
	25	142	88	54	62	38
	26	90	60	30	67	33
	27	54	43	11	80	20
	28	30	29	1	97	3
	29	11	11	0	100	0
	30	1	1	0	100	0
	31	0	0	0	0	0

Figure 8 Sheet 5

The following parameter cards must be supplied for each program run:

Symbolic Location and Contents			Card Number*	Parameter Definition	Parameter Limits
WMIN	DEC	W_{min}	3790	Minimum square width to be considered	1 to W_{max}
WMAX	DEC	W_{max}	3800	Maximum square width to be considered	1 to P_{wd}
PLC	DEC	P_{lc}	3792	Picture length (number of lines per picture)	1 to magnetic tape storage limit
WP	DEC	W_p	3834	Number of pictures	1 to magnetic tape storage limit
PICWID	DEC	P_{wd}	3819	Picture width (number of picture elements per line)	1 to 1000 (providing number of different regions per line does not exceed 60)

3.4 Logical description

One line of picture elements is read in from magnetic tape at a time. The program assigns a column height to each element of the line as follows. The column heights of elements within an interval are incremented by 1; the column heights of elements not in an interval are reset to zero. Initially all column heights are set to zero.

Each interval is processed as follows: The region number of each interval (a, b) is examined to see whether it is in a previously processed region. If not, the region number, row number, and column

*As shown on the assembly listing of the program (Figure 10) number are placed in an empty line of the Region Table. A flag, called the usage indicator, which is explained later on, is set for this region.

Each element within the current interval is examined to see whether it will complete a square of the width being considered (the current width). A square width counter is reset to zero at the beginning of the interval. For each element of the interval, processing proceeds as follows: If the column height is found equal to or greater than the current width, the current square is "high" enough, and therefore the square width counter is incremented by 1. If now, the square width counter itself equals or exceeds the current width, the current square is also "wide" enough, and therefore the frequency of squares of the width being examined for the region in process is incremented by 1.

If the column height is too small, the square width counter is reset to zero. (The square width counter thus represents the number of consecutive columns which are "high" enough.) This is done for all square widths from WMIN to WMAX by setting the current width first to WMIN and incrementing it by 1 up to and including WMAX. However, if the frequency of the squares of some width W less than WMAX is found to be zero, the square sizes W + 1, ..., WMAX are not considered. Succeeding intervals are processed in the same fashion.

Important to the program are the usage indicators referred to above. There is one usage indicator for each line of the Region Table, which holds information about regions currently in process. Initially all the usage indicators are set to zero. If the region number for an interval matches none in the Region Table, this indicates a new region to be processed. A new entry is made for this region into the first empty line of the table, i.e., the first line whose usage indicator is zero. This usage indicator is then reset to 1, indicating that this region is now in process. If a region of the interval has been processed in the preceding line (i.e., if the region number matches one already in the table), the usage indicator for that region in the Region Table is reset to 1, indicating that this region is still in process.

When all the intervals of a line have been processed, each usage indicator is examined. If a usage indicator is 1, that line of the Region Table was in process in the present cycle (one picture line is processed each cycle); that usage indicator is reset to minus 1. If the usage indicator is initially minus 1, that line of the region was in process in the previous cycle but not in the present cycle. Therefore, this region is "closed out," and its data are ready for output. This usage indicator is reset to zero (indicating that its line in the Region Table can be reused subsequently), and data for that region are then output.

When outputting a line of the Region Table, the frequency of exterior and interior squares and the percentage of exterior and interior squares are obtained using the fact that the frequency of interior squares of width t is equal to the frequency of all squares of width (t + 2). The frequency of exterior squares of width t is then equal to the frequency of total squares of width t minus the frequency of the interior squares of width t.

A flow chart and symbolic program listing for REST are shown as Figures 9 and 10.

Figure 9

REST Flow Chart



COLM	Column Height Table
REGTB	Region Number Table
FRECT	Square Frequency Table
USAGE	Usage Indicators

(WMAX)	Maximum Square Size
(WMIN)	Minimum Square Size

Initialize Picture Line Counter (PLV)

Have all lines of picture been processed?
(PLC) = Number of lines in picture
VC = Variable Connector

Increment picture line counter by 1 (PLV)

Is number of intervals (NUOIN) on this line = 0? If so, go to part of program to clear out column heights for this line.

Set left boundary of gap before first interval = 0

Figure 9 - Sheet 1



Ri = Region Number of Interval
Ai = Left End Point of Interval
Bi = Right End Point of Interval

(PRERIN) = Number of elements in gap between intervals

Is (LIN) left end point of interval first element in line?

Hc = Column Count of Left End Point of Gap
Ha = Column Count of Left End Point of Interval

Is current interval in a previously processed region?

Search for first empty line of Region Table where USAGE INDICATOR = 0

(REGIO) = i^{th} line of Region Number Table
(PLV) = i^{th} line of Row Number Table
(LIN) = i^{th} line of Column Number Table

Figure 9 - Sheet 2



Store Line Number of Region Table being processed

Set Usage Indicator of i^{th} line of Region Table = 1

(SIZE) = Number of elements in interval being processed

(WMAXP) = Maximum Square Size to be considered plus 2

Set Frequency Table for line number of Region Table being processed

Increment Column Heights in interval being processed by 1

Set Minimum Square Size (WMIN) to Current Square Size (WCUR)

Set Switch (which determines whether a square has been counted) = 0

Set Square Width Counter (J) = 0

Set Column Counter for Interval (K) = 0

Figure 9 - Sheet 3



Is Square Height large enough ?

Increment Square Width Counter by 1

Is Square Width large enough?

Add 1 to Square Frequency WCUR in Frequency Table

Set Switch = 1 to indicate one square counted this cycle of size (WCUR)

Set Square Width (J) Counter = 0

Figure 9 - Sheet 4



Have all columns for current Interval been processed for this square size?



Has square of size (WCUR) been found for present interval?

Has Maximum Square Size been processed for cycle?

Set Right Side of interval = to left boundary of gap

Figure 9 - Sheet 5



Increase Current Square Size by 1

Is right endpoint of last interval in last column of line?
(PICWID) = width of picture

Set all column heights to right of last interval = 0

Figure 9 - Sheet 6



Set counter for testing usage indicators = 0

Is i^{th} line of Region Table empty?

Was i^{th} line of Region Table in process this line?

Set Usage Indicator of this line of Region Table = 0

Set Usage Indicator of this line of Region Table = - 1

Have all lines of Region Table been tested?

Increment counter for testing usage indicators by 1

Figure 9 - Sheet 7



Prepare Row Number and Column Number for outputting

Set up counter for number of square sizes

Set frequency of interior squares of size i-1 = to frequency of total squares of two sizes larger

Set frequency of exterior squares of size i-1 = to frequency of total squares of size i-1 - frequency of interior squares of size i - 1

Is frequency of squares of this size equal to zero?

Figure 9 - Sheet 8



Compute Percentage of Exterior and Interior Squares

Set both percentage of exterior squares and percentage of interior squares = 0

Set up counter for number of square sizes

Set DIMENS equal to minimum square size

Figure 9 - Sheet 9



Set up square frequencies and square percentages for outputting

Set square frequencies of this line of Region Table = 0

Figure 9 - Sheet 10



Set all column heights equal to zero

Figure 9 - Sheet 11

Figure 10

REST Symbolic Program Listing

```
*                REST   REGION ENCLOSED SQUARE TABULATOR
NEWPCT LXA       REGNS,1
       LXA       REGNS,2
       TXI       *+1,2,USAGE
       SXA       SC3BBB,2
       SXA       SCANN,2
       SXA       USE,2
       SXA       SC10A-2,2
       SXA       SC10+1,2
       SXA       SC10A-5,2
       SXA       SC10A+1,2
USE    STZ       **,1          SET USAGE INDICATOR FOR EACH LINE OF TABLE=0
       TIX       USE,1,1
       LXA       REGNS,4       NUMBER OF REGIONS PLUS ONE INTO IR4
       LXA       REGNS,1       NUMBER OF REGIONS PLUS ONE INTO IR1
       TXI       *+1,1,REGTB
       SXA       SC10+3,1      SET UP ADDRESS FOR OUTPUTTED REGION NUMBER
       SXA       SCAN+1,1
       SXA       BACK,1
BACK   STZ       **,4          SET ALL REGION NUMBERS IN TABLE EQUAL TO ZERO
       TIX       BACK,4,1
       AXT       7200,2
       AXT       7200,4
       TXI       *+1,4,FRECT
       SXA       FRETA,4
FRETA  STZ       **,2          CLEAR ALL FREQUENCIES
       TIX       FRETA,2,1
       LXA       PICWID,4      PICTURE WIDTH INTO IR4
       LXA       PICWID,1      PICTURE WIDTH INTO IR1
       TXI       *+1,1,COLM
       SXA       LAST+2,1
       SXA       COLUMN,1
COLUMN STZ       **,4          SET ALL COLUMN HEIGHTS EQUAL TO ZERO
       TIX       COLUMN,4,1
SC0    CAL       =6B17
       CALL      (STH)                HEADINGS
       PZE       FORM,0,1
       AXT       20,1
D4     LDQ       FHEAD+20,1
       STR
       TIX       D4,1,1
       AXT       20,1
D5     LDQ       HEAD+20,1
       STR
       TIX       D5,1,1
       CALL      (FIL)
       CLA       WMAX          MAXIMUM SQUARE SIZE TO BE CONSIDERED
       SUB       WMIN          MINIMUM SQUARE SIZE TO BE CONSIDERED
       ADD       ONE
       STO       RANGE         NUMBER OF SQUARE SIZES
SC1    STZ       PLV           INITIALIZE PICTURE LINE COUNTER
SC1A   CLA       PLC           ANY MORE LINES OF
       SUB       PLV                  PICTURE TO PROCESS.
```

Figure 10 - Sheet 1

```
VARING TZE      VC
       CLA      PLV                OBTAIN NEXT LINE
       ADD      ONE                PLV EQUALS PLV PLUS
       STO      PLV                ONE
       CAL      =9B17              READ IN NEXT LINE
       CALL     (TSB)             OF PICTURE
       STR
       STQ      NUOIN
       CLA      NUOIN        NUMBER OF INTERVALS
       TZE      SC1E
       LXA      NUOIN,2      NUMBER OF INTERVALS INTO IR2
       LXA      NUOIN,4      NUMBER OF INTERVALS INTO IR4
       TXI      *+1,4,INTER
       SXA      SC1C,4
SC1B   SXA      SC1D,2
       STR
       STQ      INTREV
       CLA      INTREV
SC1C   STO      **,2
SC1D   AXT      **,2
       TIX      SC1B,2,1
SC1E   CALL     (RLR)        READ IN PRESENT LINE
       CLA      NUOIN        NUMBER OF INTERVALS
       TZE      LAST
       LXA      NUOIN,2      NUMBER OF INTERVALS INTO IR2
SC2A   LXA      NUOIN,4      NUMBER OF INTERVALS INTO IR4
       TXI      *+1,4,INTER
       SXA      SC2D,4
       SXA      SC3AA,4
       SXA      SC2BB,4
       SXA      SC2BBB,4
       SXA      SC3,4
       SXA      SC3AAA,4
       SXA      INTEG,4
       STZ      PRERIN       INITIALIZE RIGHT SIDE OF PREVIOUS INTERVAL
SC2B   CAL      MASK3
INTEG  ANA      **,2               OBTAIN RIGHT MEMBER OF INTERVAL
       STO      RIN
SC2BB  CAL      **,2
       ARS      12
SC2BBB STO      **,2
       CAL      MASK3
SC2D   ANA      **,2
       STO      LIN          LEFT SIDE OF INTERVAL
       CLA      LIN          LEFT SIDE OF INTERVAL
       SUB      PRERIN       RIGHT SIDE OF PREVIOUS INTERVAL
       SUB      ONE
       STO      PRERIN       GAP BETWEEN INTERVALS
       LXA      PRERIN,4     INTO IR4
       LXA      LIN,1        LEFT SIDE OF INTERVAL INTO IR1
       TXI      *+1,1,COLM
       TXI      *+1,1,-1
       SXA      ZEROOS,1
```

Figure 10 - Sheet 2

```
        SXD     SC3C,1
        SXA     SC7,1
        SXA     SC8C,1
        CLA     LIN             LEFT SIDE OF INTERVAL
        SUB     ONE
        TZE     SC3
ZEROOS  STZ     **,4            SET ALL COLUMN HEIGHTS BETWEEN LAST
        TIX     ZEROOS,4,1      INTERVAL AND PRESENT INTERVAL EQUAL TO ZERO
SC3     CAL     **,2
        ARS     12
SC3AAA  STO     **,2
        CAL     MASK3
SC3AA   ANA     **,2
        STO     REGIO           REGION NUMBER OF INTERVAL
        LXA     REGNS,1
        TXI     *+1,1,REGTB
        SXA     ADDRE,1
        LXA     REGNS,4
SCAN    CLA     REGIO           IS CURRENT INTERVAL IN
        SUB     **,4            A PREVIOUSLY PROCESSED REGION
        TZE     SC3B
        TIX     SCAN,4,1
        LXA     REGNS,4
SCANN   CLA     **,4            SEARCH FOR FIRST EMPTY LINE OF REGION TABLE
        TZE     OK
        TIX     SCANN,4,1
OK      PXA     ,4
        STO     LINE
        CLA     REGIO           PLACE REGION NUMBER OF
ADDRE   STO     **,4            CURRENT INTERVAL IN REGION TABLE
        LXA     REGNS,1         NUMBER OF REGIONS INTO IR1
        TXI     *+1,1,ROW
        SXA     SC11A,1         SET UP ADDRESS FOR OUTPUTTED COLUMN NUMBER
        SXA     ADDRF,1
        CLA     PLV             ROW NUMBER OF CURRENT LINE
ADDRF   STO     **,4
        LXA     REGNS,1         NUMBER OF REGIONS INTO IR1
        TXI     *+1,1,COL
        SXA     SC11B-2,1       SET UP ADDRESS FOR OUTPUTTED COLUMN NUMBER
        SXA     ADDRG,1
        CLA     LIN             LEFT SIDE OF INTERVAL
ADDRG   STO     **,4
SC3B    PXA     ,4
        STO     LINE
        CLA     SIXTY1
        SUB     LINE
        STO     LINE
        CLA     ONE
SC3BBB  STO     **,4
        CLA     RIN             RIGHT SIDE OF INTERVAL
        SUB     LIN             LEFT  SIDE OF INTERVAL
        ADD     ONE
```

Figure 10 - Sheet 3

```
        STO     SIZE          SIZE OF INTERVAL PLUS ONE
        LXA     SIZE,1
SC3C    TXI     *+1,1,**      UPDATE COLUMN COUNTER
        SXA     SIZEY,1             COUNTER
        SXA     SIZEY+2,1
        CLA     WMAX          MAXIMUM SQUARE SIZE BEING CONSIDERED
        ADD     TWO
        STO     WMAXP
        LDQ     LINE
        MPY     WMAXP
        XCA
        STO     ADDER
        LXA     ADDER,1
        TXI     *+1,1,FRECT
        SXA     NEXT,1
        SXA     NEXT+2,1
        LXA     SIZE,4
SIZEY   CLA     **,4               ADD ONE TO COLUMN
        ADD     ONE                  HEIGHTS IN THIS INTERVAL
        STO     **,4
        TIX     SIZEY,4,1
SC5     CLA     WMIN                 SET CURRENT WIDTH TO
        STO     WCUR                 MINIMUM COUNT WIDTH
        CLA     WMAX          MAXIMUM SQUARE SIZE TO BE CONSIDERED
        SUB     WMIN          MINIMUM SQUARE SIZE TO BE CONSIDERED
        ADD     =3
        STO     MRANGE
        LXA     MRANGE,4
SC6     STZ     SWCH          SET SWCH EQUAL TO ZERO
        STZ     J             SET SQUARE WIDTH COUNTER EQUAL TO ZERO
        STZ     K             SET COLUMN COUNTER FOR THIS INTERVAL =0
SC7     CLA     **                     IS SQUARE HEIGHT
        SUB     WCUR                 LARGE ENOUGH.
        TMI     SC7A
        CLA     J
        ADD     ONE
        STO     J                    IS SQUARE WIDTH
        SUB     WCUR                 LARGE ENOUGH.
        TMI     SC7B
NEXT    CLA     **,4                 ADD 1 TO FREQUENCY COUNT
        ADD     ONE                  OF SQUARE OF SIZE WCUR,
        STO     **,4                 REGION R.
        CLA     ONE                  AT LEAST ONE SQUARE
        STO     SWCH                 COUNTED THIS CYCLE.
        TRA     SC7B
SC7A    CLA     ZERO
        STO     J
SC7B    CLA     K             COLUMN COUNTER FOR INTERVAL BEING PROCESSED
        ADD     ONE
        STO     K             COLUMN COUNTER FOR INTERVAL BEING PROCESSED
SC8     CLA     K                    IS IT END OF
        SUB     SIZE                 CURRENT INTERVAL.
```

Figure 10 - Sheet 4

```
        TZE     SC8A
        LXA     SC7,1         GO TO NEXT
        TXI     *+1,1,1       COLUMN FOR
        SXA     SC7,1         COUNTING
        TRA     SC7
SC8A    CLA     SWCH                AT LEAST ONE SQUARE
        SUB     ONE                 COUNTED THIS CYCLE.
        TNZ     *+2
        TIX     SC8B,4,1            LARGEST SQUARE SIZE
        CLA     RIN           RIGHT SIDE OF INTERVAL
        STO     PRERIN        RIGHT SIDE OF PREVIOUS INTERVAL
        TIX     SC2B,2,1      ANY MORE INTERVALS THIS LINE
        TRA     SC9
SC8B    CLA     WCUR
        ADD     ONE
        STO     WCUR
SC8C    AXT     **,1
        SXA     SC7,1
        TRA     SC6
SC9     CLA     RIN           RIGHT SIDE OF INTERVAL
        ADD     ONE
        SUB     PICWID        WIDTH OF PICTURE
        TPL     SC9B
        CLA     PICWID
        SUB     RIN           RIGHT SIDE OF INTERVAL
        STO     ENDING        LENGTH OF LAST GAP IN INTERVAL
        LXA     ENDING,4      LENGTH OF LAST GAP IN INTERVAL INTO IR4
        LXA     PICWID,1      PICTURE WIDTH INTO IR1
        TXI     *+1,1,COLM
        SXA     ZEROO,1
ZEROO   STZ     **,4          SET ALL COLUMN HEIGHTS TO RIGHT OFINTERVAL=0
        TIX     ZEROO,4,1           EQUAL TO ZERO
SC9B    LXA     REGNS,1
SC10    CLA     ZERO                IS THIS LINE OF
        SUB     **,1                REGION TABLE EMPTY.
        TZE     SC11
        CLA     **,1
        STO     REGNC               REGION BEING CONSIDERED
        CLA     ONE                 WAS THERE AN INTERVAL IN
        SUB     **,1                REGION R THIS CYCLE.
        TZE     SC10A
        SXA     SAVE,1
        STZ     **,1          SET USAGE INDICATOR EQUAL TO ZERO
        TRA     SC11A
SC10A   CLS     ONE                 SET USAGE INDICATOR EQUAL
        STO     **,1                TO MINUS ONE
SC11    TIX     SC10,1,1
        TRA     SC1A
SC11A   CLA     **,1          ROW NUMBER OF REGION BEING OUTPUTTED
        STO     ROWC
        CLA     **,1
        STO     COLC
```

Figure 10 - Sheet 5

```
SC11B   PXA     ,1
        STO     LINE
        CLA     SIXTY1
        SUB     LINE
        STO     LINE
        LDQ     LINE
        MPY     WMAXP
        XCA
        STO     ADDER
        LXA     ADDER,4
        TXI     *+1,4,FRECT
        SXA     BEGIN,4
        TXI     *+1,4,-2
        SXA     SC21,4
        SXA     ORIGIN,4
        SXA     DIVIDE,4
        SXA     BEGIN+2,4
        LXA     RANGE,2      NUMBER OF SQUARE SIZES INTO IR2
BEGIN   CLA     **,2
        STO     INT,2        FREQUENCY OF INTERIOR SQUARES
        CLA     **,2                   OF INTERIOR SQUARES OF SIZE
        SUB     INT,2        FREQUENCY OF INTERIOR SQUARES
        STO     EXT,2        FREQUENCY OF EXTERIOR SQUARES
        TIX     BEGIN,2,1
        LXA     RANGE,2      NUMBER OF SQUARE SIZES INTO IR2
ORIGIN  CLA     **,2                   IS FREQUENCY OF SQUARES
        TZE     ZEROIN                 EQUAL TO ZERO.
        CLA     INT,2      FREQUENCY OF INTERIOR SQUARES
        TZE     SC17
        XCA
        CLA     ZERO
        MPY     TWOHUN
DIVIDE  DVP     **,2
        XCA
        LRS     1
        RND
        STO     INTPE,2      PERCENTAGE OF INTERIOR SQUARES
TESTY   CLA     HUNDR
        SUB     INTPE,2      PERCENTAGE OF INTERIOR SQUARES
        STO     EXTPE,2      PERCENTAGE OF EXTERIOR SQUARES
TEST    TIX     ORIGIN,2,1
        TRA     SC19
SC17    STZ     INTPE,2
        TRA     TESTY
ZEROIN  CLA     ZERO                   SET INTERIOR AND EXTERIOR
        STO     INTPE,2      PERCENTAGE OF INTERIOR SQUARES
        STO     EXTPE,2      PERCENTAGE OF EXTERIOR SQUARES
        TRA     TEST
SC19    LXA     RANGE,2      NUMBER OF SQUARE SIZES INTO IR2
        CLA     WMIN         MINIMUM SQUARE SIZE
        STO     DIMENS       SQUARE SIZE TO BE PRINTED
SC21    CLA     **,2
        STO     TFREQ
```

Figure 10 - Sheet 6

```
         CLA     INT,2          FREQUENCY OF INTERIOR SQUARES
         STO     IFREQ
         CLA     EXT,2          FREQUENCY OF EXTERIOR SQUARES
         STO     EFREQ
         CLA     EXTPE,2        PERCENTAGE OF EXTERIOR SQUARES
         STO     EPERC
         CLA     INTPE,2        PERCENTAGE OF INTERIOR SQUARES
         STO     IPERC
         SXA     SC20,2
SCOTPT   CAL     =6B17
         CALL    (STH)
         PZE     FFORM,0,1
FIRST    LDQ     REGNC                 PRINT REGION NUMBER
         STR
         LDQ     ROWC                  PRINT ROW NUMBER OF
         STR                           REGION.
         LDQ     COLC                  PRINT COLUMN NUMBER OF
         STR                           REGION.
         LDQ     DIMENS         PRINT SQUARE SIZE
         STR
         LDQ     TFREQ          PRINT TOTAL NUMBER OF SQUARES
         STR                           SQUARES.
         LDQ     EFREQ          PRINT NUMBER OF EXTERIOR SQUARES
         STR                           SQUARES.
         LDQ     IFREQ          PRINT NUMBER OF INTERIOR SQUARES
         STR                           SQUARES.
         LDQ     EPERC          PRINT PERCENTAGE OF EXTERIOR SQUARES
         STR                           EXTERNAL SQUARES.
         LDQ     IPERC          PRINT PERCENTAGE OF INTERIOR SQUARES
         STR                           INTERNAL SQUARES
         CALL    (FIL)
         CLA     TFREQ
         TZE     FINISH
         CLA     DIMENS         SQUARE SIZE TO BE PRINTED
         ADD     ONE
         STO     DIMENS         SQUARE SIZE TO BE PRINTED
SC20     AXT     **,2
         TIX     SC21,2,1       HAVE ALL SQUARE SIZES BEEN OUTPUTTED
FINISH   CAL     =6B17
         CALL    (STH)
         PZE     FORMAT,0,1
         LDQ     SKIPLI
         STR
         CALL    (FIL)
SAVE     AXT     **,1
         PXA     ,1
         STO     LINE
         CLA     SIXTY1
         SUB     LINE
         STO     LINE
         LDQ     LINE
         MPY     WMAXP
```

Figure 10 - Sheet 7

```
        XCA
        STO     ADDER
        LXA     ADDER,4
        TXI     *+1,4,FRECT
        SXA     CLEAR,4
        LXA     MRANGE,4
        CLA     ZERO
CLEAR   STO     **,4
        TIX     *-2,4,1
        TRA     SC11
LAST    LXA     PICWID,2      PICTURE WIDTH INTO IR2
        CLA     ZERO
        STO     **,2
        TIX     *-2,2,1
        TRA     SC9B
FORM    BCI     5,(20A6/20A6///)
FHEAD   BCI     5,1
        BCI     5,
        BCI     5, SQUARE FREQUENCIES
        BCI     5, SQUARE PERCENTAGES
HEAD    BCI     5,  REGION NUMBER ROW COLUMN
        BCI     5, SQUARE SIDE LENGTH
        BCI     5, TOTAL EXTERIOR INTERIOR
        BCI     5, EXTERIOR INTERIOR
FFORM   BCI     9,(1J13,1J6,1J7,1J22,1J17,1J9,1J9,1J15,1J9/)
FORMAT  BCI     1,(1A6/)
SKIPLI  BCI     1,
USAGE   BSS     60
RIN     BSS     1
LIN     BSS     1
INTREV  BSS     1
NUOIN   BSS     1
REGIO   BSS     1
COLM    BSS     200
SWCH    BSS     1
J       BSS     1
PRERIN  BSS     1
WCUR    BSS     1
ROWC    BSS     1
COLC    BSS     1
REGNC   BSS     1
FRECT   BSS     7200
K       BSS     1
SIZE    BSS     1
INT     BES     120           NUMBER OF INTERIOR SQUARES
EXT     BES     120           NUMBER OF EXTERIOR SQUARES
INTPE   BES     120           PERCENTAGE OF INTERIOR SQUARES
EXTPE   BES     120           PERCENTAGE OF EXTERIOR SQUARES
ROW     BSS     60
COL     BSS     60
PLV     BSS     1
```

Figure 10 - Sheet 8

```
INTER   BSS    60
REGTB   BSS    60
ADDER   BSS    1
ZERO    DEC    0
ONE     DEC    1
TWO     DEC    2
RANGE   BSS    1              NUMBER OF SQUARE SIZES
WMIN    DEC    2
PLC     DEC    4125           NUMBER OF LINES
MASK1   OCT    777700000000
MASK2   OCT    000077770000
MASK3   OCT    000000007777
CSTOP   PZE    STOP
CVC     PZE    VC
WMAX    DEC    120            MAXIMUM SQUARE SIZE TO BE CONSIDERED
REGNS   DEC    60
ENDING  BSS    1              LENGTH OF LAST GAP IN INTERVAL
PICWID  DEC    120            WIDTH OF PICTURE
HUNDR   DEC    100
TWOHUN  DEC    200
MRANGE  BSS    1
DIMENS  BSS    1
TFREQ   BSS    1
IFREQ   BSS    1
EFREQ   BSS    1
IPERC   BSS    1
EPERC   BSS    1
WTMAX   DEC    120
WP      DEC    1              NUMBER OF PICTURES
LINE    BSS    1
WMAXP   BSS    1
SIXTY1  DEC    61
VC      CLA    CSTOP
        STA    VARING
        TRA    SC9B
NEWPIC  CLA    CVC
        STA    VARING
        TRA    NEWPCT
STOP    CLA    WP             NUMBER OF PICTURES
        SUB    ONE
        STO    WP             NUMBER OF PICTURES
        TNZ    NEWPIC
```

Figure 10 - Sheet 9

4. Application to TIROS pictures and nephanalyses

The RAMP and REST programs have been applied to two types of pictorial material:

(a) A set of digitized TIROS cloud cover pictures

(b) A set of outline-map nephanalyses made from TIROS pictures

4.1 TIROS pictures

Arking (1964) and others have demonstrated the possibility of discriminating cloud from noncloud on TIROS pictures with fair reliability by using an appropriate brightness threshold. He has used this approach in a study of percentage cloud cover as a function of season and geographic location. If RAMP is applied to TIROS pictures, a further analysis of the cloud cover becomes possible in terms of numbers and areas of large connected cloud masses (overcast) as well as total area and "piece size spectrum" of broken cloud cover. Application of REST then provides information about the shapes of the connected cloud masses.

For direct cloud picture input to RAMP a magnetic tape was used containing digitized versions of ten pictures transmitted by TIROS V. Figure 1 is a reproduction of a high-speed chemical printer output of the first picture of this sequence. (The cooperation of Dr. Albert Arking of the NASA Institute for Space Studies, New York, in making this tape available is gratefully acknowledged.)

The format of this tape is as follows. The first record is an identifying Control Record in FORTRAN binary format containing the following words of information:

(1) FORTRAN control word

(2) Output tape reel number

(3) TIROS number

(4) Integer 1

(5) Number of frames this sequence

(6) R/O pass number

(7) Sequence number

(8) Same as (5)

The next record is a Sequence-Data Record in FORTRAN binary format containing the following words of information:

(1) FORTRAN control word

(2) TIROS number

(3) R.O. number

(4) Mode

(5) Camera

(6) Sequence position

(7) Number of sequences

(8) and (9) Date

(10) Number of days

(11), (12), and (13) Time

(14) Number of frames

(15) Usable frames

(16) ID scheme

(17) and (18) Latitude and longitude

(19) Height

(20) Right ascension

(21) Declination

(22), (23), (24) P.P. longitude, latitude and elevation

(25) Elevation of sub-satellite point

(26) Nadir angle

(27) P.S. angle

(28) Ersatz

(29) Roll angle

(30) Time (minutes)

(31) ID number

(32) Frame number

Following the Sequence-Data Record are twenty-four 10-line Gray Level Records in non-FORTRAN binary format. Picture elements are packed six per computer word, six bits per element. Brightness levels range from 63 (the brightest) to 0 (the darkest). Each of these records is 390 words in length. The picture thus consists of 240 lines of 234 elements each.*

*This is a 1/4 area scale reduction of the data tape format of 480 lines of 468 elements each, as transmitted directly from the TIROS satellite.

Following this is an eight-word End-of-Line Record marking the end of the first frame. Each of the other nine frames on the tape similarly contains a Sequence-Data Record, 24 Gray Level Records, and an End-of-Line Record.

Only one format change was necessary for input of these pictures to RAMP: the 10-line Gray Level Records were broken up into ten records of one line (234 picture elements) each.

RAMP outputs at eight different brightness threshold levels for the TIROS picture of Figure 1 are shown as the left-hand pictures in Figure 11. Element digital values range from 0 (darkest) to 63 (brightest). The thresholds used were 8, 12, 16, 20, 24, 28, 32, and 36. In each example, elements which are brighter (greater in digital value) than the threshold are printed out; elements darker than threshold are left blank. The right-hand pictures in Figure 11 show the same set of pictures "reversed." Elements whose digital values are less than or equal to the brightness threshold are printed out, while elements exceeding the threshold are left blank. The same set of thresholds was used. An advantage of the reversal is that the bright clouds now correctly appear on the printout as white patches against a dark printed background, so that the printout now resembles a "positive" picture.

4.2 Nephanalyses

The application of RAMP and REST to cloud cover maps, or nephanalyses, provides information about the sizes and shapes of the regions which contain single types of cloud cover as defined by the maps.

Figure 11

RAMP Output for a Typical Cloud Picture

at Various Threshold Levels



FIGURE 11 SHEET 1



FIGURE II SHEET 2



FIGURE II SHEET 3



FIGURE II SHEET 4



FIGURE 11 SHEET 5



FIGURE II SHEET 6



FIGURE II SHEET 7



FIGURE II SHEET 8

Besides this climatological map analysis application, RAMP and REST have also been used to establish basic parameters for a proposed approach to automatic cloud cover mapping directly from TIROS pictures. This approach involves the analysis of a set of square neighborhoods on a picture and the identification of the cloud cover contained in these neighborhoods. Clearly, the cloud cover regions identified by such a process cannot be smaller than the square neighborhoods which are analyzed. Conversely, a study of the sizes of squares which will fit into the regions on a cloud cover map provides an upper bound to the neighborhood sizes which can be used if a given proportion of the regions on an average map are to be identified.

The typical maps of Figure 2 were scanned as described in Section 2.2 (Figure 2 illustrates the possibility of putting many pictures on a single digital tape. An upper bound to the number of such pictures is imposed by the requirement that the map lines be wide enough to remain free of gaps after digitizing.) RAMP outputs for two of these maps were shown as Figures 3 and 7; the REST output for one of them, as Figure 8.

References

Arking, A. Latitudinal Distribution of Cloud Cover from TIROS III Photographs, Science 143, 1964, 569-572

Dinneen, G. P. Programming Pattern Recognition, Proceedings, 1955 Western Joint Computer Conference, 94-100

Freeman, H. Techniques for the Digital Computer Analysis of Chain-Encoded Arbitrary Plane Curves, Proceedings, National Electronics Conference 17, 1961, 421-432

Hodes, L. Machine Processing of Line Drawings, MIT Lincoln Laboratory Report 54 G-0028, 7 March 1961

Kirsch, R. A. et al Experiments in Processing Pictorial Information with a Digital Computer, Proceedings, 1957 Eastern Joint Computer Conference, 221-229

Kirsch, R. A. Computer Interpretation of English Text and Picture Patterns, National Bureau of Standards Report 8394, 1 April 1964

Narasimhan, R. Labeling Schemata and Syntatic Descriptions of Pictures, Information and Control 7, 1964, 151-179

Sutherland, I. E. Sketchpad: A Man-Machine Graphical Communication System, MIT Lincoln Laboratory Technical Report No. 296, 30 January 1963

Unger, S. H. A Computer Oriented Toward Spatial Problems, Proceedings of the IRE 46, 1958, 1744-1750